Pluralize

A Smooth CoffeeScript example by autotelicum

How do you add 's' to a word when there is more than one of something?

It is often seen in JavaScript with the ternary operator ?: but that is hard-coding at the expense of future maintenance and internationalization.

First some preliminaries. Let's show the results whether we are running in a server or a web client. And wrap the different methods in a test function so we can see if they work.

```
show = if exports? then console.log else alert
test = (points) ->
```

Readable ways

The simplest readable way

```
show "0: You got #{points} point#{if points > 1 then 's' else ''}"
```

Encapsulate in a function, future i18n will be easier

```
pluralIf = (stem, cond) -> stem + (if cond then 's' else '')
show "1: You got #{points} #{pluralIf 'point', points > 1}"
```

Encapsulate in a function, handle negative numbers

```
pluralUnless = (stem, cond) -> stem + (unless cond then 's' else '')
show "2: You got #{points} #{pluralUnless 'point', -2 < points < 2}"</pre>
```

Encapsulate in a String method, could collide with another definition

```
String::pluralIf ?= (cond) -> this + (if cond then 's' else '')
show "3: You got #{points} #{'point'.pluralIf points > 1}"
```

Tricks

Use the existential operator to catch the singular undefined

```
show "4: You got #{points} point#{('s' if points > 1) ? ''}"
```

Convert the condition to a number and do an array lookup

```
show "5: You got #{points} point#{['','s'][+(points > 1)]}"
```

Use the inverted condition to ask for a letter.

```
true => charAt 1 => '' — false => charAt 0 => 's'
show "6: You got #{points} point#{'s'.charAt points <= 1}"</pre>
```

Create an array whose length is determined by the condition. It consists of empty elements so join it with 'c'

```
show "7: You got #{points} point#{Array(1+(points>1)).join 's'}"
```

Concatenate to an empty array, undefined does nothing

```
show "8: You got #{points} point#{[].concat ('s' if points > 1)}"
```

Contributed

Create an array with either 's' or undefined — from satyr

```
show "9: You got #{points} point#{['s' if points > 1]}"
```

Test

Run a test for a small range of numbers

```
test n for n in [-3..3]
```

Formats CoffeeScript Markdown PDF HTML

Copyright autotelicum © 2554/2011 (CC) BY-SA